

ŠOTKIJEVA DIODA

- TEHNOLOŠKI NIZ I ELEKTRIČNE
KARAKTERISTIKE -

ATLAS

1. Definisiranje mreže: **mesh**

Tehnološki niz (struktura) komponente, takođe je moguće simulirati korišćenjem Atlas-a.

```
go atlas
```

```
mesh space.mult=1.0
```

```
x.mesh loc=0.00 spac=0.5  
x.mesh loc=3.00 spac=0.2  
x.mesh loc=5.00 spac=0.25  
x.mesh loc=7.00 spac=0.25  
x.mesh loc=9.00 spac=0.2  
x.mesh loc=12.00 spac=0.5
```

```
y.mesh loc=0.00 spac=0.1  
y.mesh loc=1.00 spac=0.1  
y.mesh loc=2.00 spac=0.2  
y.mesh loc=5.00 spac=0.4
```

mesh - naredba za definisanje mreže

space.mult - faktor skaliranja (odnosi se na spacinge u mreži)

x.mesh/y.mesh - definisanje mreže po x/y osi

loc (location) - definisanje koordinata

spac (spacing) - definisanje razmaka u mreži (definisanje gustine mreže)

2. Definisanje regiona: **region**

```
region num=1 silicon
```

- **Definisanje regiona:**

region - specificiranje lokacije materijala u prethodno definisanoj mreži;
- parametar **num** označava broj regiona.

3. Lociranje elektroda: **electrode**

```
electrode name=anode x.min=5 length=2  
electrode name=cathode bottom
```

- **Lociranje elektroda:**

electrode - specificiranje lokacija i naziva elektroda u prethodno definisanoj mreži.

4. Distribucija dopiranja: **doping**

```
doping n.type conc=5e16 uniform ← n- epi doping: formiranje silicijumskog
                                supstrata n tipa

doping p.type conc=1e19 x.min=0 x.max=3 junc=1 rat=0.6 gauss ← Guard-ring doping:
doping p.type conc=1e19 x.min=9 x.max=12 junc=1 rat=0.6 gauss ← formiranje zaštitnih
                                                                prstenova p tipa

doping n.type conc=1e20 x.min=0 x.max=12 y.top=2 y.bottom=5 uniform ← n+ doping
```

```
save outfile=sotki.str
tonyplot sotki.str
```




Čuvanje i prikaz podataka, u Atlasu se podaci čuvaju naredbom `save`, a ne naredbom `structure`.

- Distribucijom dopiranja specificiraju se profili dopiranja analitički ili iz ulaznog fajla. Analitički profili specificiraju kako će Atlas generisati profile dopiranja iz analitičkih funkcija:

1. **Gausov**- koristi Gausovu analitičku funkciju, definiše se `n.type` ili `p.type` i jedna od sledećih grupa specifikacije profila: koncentracija i dubina spoja, doza i karakterizacija, koncentracija i karakterizacija;
2. **uniforman**- koristi uniformnu (konstantnu) analitičku funkciju, definiše se `n.type` ili `p.type` i parametar koncentracije.

DIODE - Šotkijeva dioda

Na slici je prikazana dobijena 2D struktura Šotkijeve diode n-tipa sa jako dopiranim oblastima zaštitnog prstena (guard-ring) p-tipa na levoj i desnoj strani strukture. Anoda diode je na vrhu komponente, a jako dopirana katoda na dnu komponente.



5. Definisanje modela i kontakta: **model** i **contact**

```
model conmob fldmob srh auger bgn
contact name=anode workf=4.97
```



- Definisanje modela:
 - **conmob**- model zavisnosti pokretljivosti od koncentracije nosilaca;
 - **fldmob**- model zavisnosti pokretljivosti od električnog polja;
 - **srh**, **auger**- SRH i Auger modeli rekombinacije;
 - **bgn**- model sužavanja širine zabranjene zone.

Ključna naredba za podešavanje Šotkijevog kontakta je naredba. **contact**- koristi se za definisanje **work function**-a za elektrodu ($wf=4.97$ obezbeđuje visinu Šotkijeve barijere od 0.8 V). Podrazumevana visina barijere je 0 - idealni omski kontakt (u ovom slučaju katoda je idealni omski kontakt).

Simulaciija električnih karakteristika

```
solve init  
method newton  
log outfile=sotki-kka.log  
solve vanode=0.05 vstep=0.05 vfinal=1 name=anode
```

```
tonyplot sotki-kka.log  
quit
```

Nije potrebno pozivati Atlas ponovo, niti definisanje modela, kontakta na koji se dovodi napon i metoda za rešavanje jednačina, jer je struktura Šotkijeve diode rađena u Atlas-u.

